```
2 - 2
```

0

Sebar Polynomials

Nicholas Wheeler
12 October 2017 (Columbus Day)

■ Introduction

I was visited on October 8--11 by Ahmed Sebbar and his wife Claudel. Our correspondence began in November 2014, when Ahmed expressed interest in my old "Applied Theta Functions" paper, but by October 2015 we were in correspondence re the "generalized Pell problem," the differential geometry of the hexenhut and certain "Sebbar polynomials." (I have not yet recovered the handwritten pages on which those polynomials were brought first to my attention.)

On the 8th, Ahmad presented me with a single handwritten page that relates again to those polynomials. And on the 11th a "New Version" of that page (different notation, more detail). It is from those pages that I work here.

■ Classical preliminaries

Chapter 22 of Abramowitz & Stegun treats Orthogonal Polynomials. Many of the generating functions (treated on pages 783-4) involve the function

$$R = \sqrt{1 - 2\,x\,h + h^2}\;;$$

which devolves from some spherical geometry in 3-space.

■ Legendre polynomials

NOTE: I produce the following mainly to make sure that certain simplified commands work.

$$\texttt{LegendreSeries = Series}\left[\frac{1}{R},\ \{h, 0, 10\}\right];$$

```
Table[{n, P_n = SeriesCoefficient[LegendreSeries, n]}, {n, 0, 10}] // TableForm
```

0	1
1	x
2	$\frac{1}{2}(-1+3x^2)$
3	$\frac{1}{2}(-3x+5x^3)$
4	$\frac{1}{8}(3-30x^2+35x^4)$
5	$\frac{1}{8}(15x-70x^3+63x^5)$
6	$\frac{1}{16}(-5+105x^2-315x^4+231x^6)$
7	$\frac{1}{16}(-35x+315x^3-693x^5+429x^7)$
8	$\frac{1}{128}(35-1260x^2+6930x^4-12\,012x^6+6435x^8)$
9	$\frac{1}{128}(315x-4620x^3+18\,018x^5-25\,740x^7+12\,155x^9)$
10	$\frac{1}{10}(-\frac{9}{8}(-\frac{7}{6}(-\frac{5}{4}(-\frac{3}{2}(-1+3x^2)+\frac{7}{3}x(-2x+\frac{5}{2}x(-1+3x^2))))+\frac{11}{5}x(-\frac{4}{3}(-2x+\frac{5}{2}x(-1+3x^2))+\frac{9}{4}x(-$

```
P10 // Simplify
```

$$\frac{1}{256}\left(-63+3465\,x^2-30\,030\,x^4+90\,090\,x^6-109\,395\,x^8+46\,189\,x^{10}\right)$$

```
Plot[{P0, P1, P2, P3, P4, P10}, {x, -1, 1}]
```



Abramowitz & Stegun (page 781) assert that P_n satisfies the following 2nd order DE

```
LDE[f_, n_] := (1 - x^2) D[f, {x, 2}] - 2 x D[f, {x, 1}] + n (n + 1) f
```

Which checks out:

```
Table[{n, Simplify[LDE[Pn, n]]}, {n, 0, 10}] // TableForm
```

0	0
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

A proof of orthogonality can be extracted from the DE. Alternatively one can evaluate

$$\int_{-1}^{1}\frac{1}{\sqrt{1-2\,x\,h+h^2}\;\sqrt{1-2\,x\,g+g^2}}\,dx$$

Mathematica struggles, produces a result subject to a long list of Assumptions—a result that after major simplification assumes the form

$$\frac{1}{\sqrt{g\,h}}\,\mathrm{Log}\left[\frac{1+\sqrt{g\,h}}{1-\sqrt{g\,h}}\right]$$

Orthogonality follows from the fact that only the product gh appears in this result:

```
Series[1/k Log[(1 + k)/(1 - k)], {k, 0, 10}] // Normal
```

$$2+\frac{2k^2}{3}+\frac{2k^4}{5}+\frac{2k^6}{7}+\frac{2k^8}{9}+\frac{2k^{10}}{11}$$

```
% /. k → Sqrt[g h]
```

$$2+\frac{2gh}{3}+\frac{2g^2h^2}{5}+\frac{2g^3h^3}{7}+\frac{2g^4h^4}{9}+\frac{2g^5h^5}{11}$$

We are led thus to the following inner product table:

```
Table[KroneckerDelta[m, n] 2/(2 n + 1), {m, 0, 5}, {n, 0, 5}] // MatrixForm
```

$$\begin{pmatrix} 2 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{2}{3} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{2}{5} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{2}{7} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{2}{9} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{2}{11} \end{pmatrix}$$

- **Gegenbauer polynomials (ultraspherical polynomials of 0th order)**

NOTE: These (as in A&S page 774) are usually denoted $C_n^{(0)}$, but I want to preserve C_n, so will denote them G_n.

The generating function is

```
-Log[R^2]
```

$-\mathrm{Log}[1+h^2-2hx]$

```
GegenbauerSeries = Series[-Log[1 + h^2 - 2 h x], {h, 0, 10}];
Table[{n, G_n = SeriesCoefficient[GegenbauerSeries, n]}, {n, 0, 10}] // TableForm
```

0	0
1	$2x$
2	$-1+2x^2$
3	$\frac{2}{3}(-3x+4x^3)$
4	$\frac{1}{2}(1-8x^2+8x^4)$
5	$\frac{2}{5}(5x-20x^3+16x^5)$
6	$\frac{1}{3}(-1+18x^2-48x^4+32x^6)$
7	$\frac{2}{7}(-7x+56x^3-112x^5+64x^7)$
8	$\frac{1}{4}(1-32x^2+160x^4-256x^6+128x^8)$
9	$\frac{2}{9}(9x-120x^3+432x^5-576x^7+256x^9)$
10	$\frac{1}{5}(-1+50x^2-400x^4+1120x^6-1280x^8+512x^{10})$

NOTE that $G_0 = 0$, so the indices begin at n = 1.

```
Plot[{G1, G2, G3, G4, G5}, {x, -1, 1}]
```




```
GDE[f_, n_] := (1 - x^2) D[f, {x, 2}] - x D[f, {x, 1}] + n^2 f
```

```
Table[{n, Simplify[GDE[Gn, n]]}, {n, 1, 10}] // TableForm
```

1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

A&S assert that the Gegenbauer polynomials are orthogonal on [-1,1] with respect to the weight function

$$w[x] = \frac{1}{\sqrt{1 - x^2}}$$

as these calculations confirm:

$$\text{Table}\left[\int_{-1}^{1} \frac{1}{\sqrt{1 - x^2}} G_m G_n \, dx, \{m, 1, 6\}, \{n, 1, 6\}\right] \text{ // MatrixForm}$$

$$\begin{pmatrix} 2\pi & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{\pi}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{2\pi}{9} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{\pi}{8} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{2\pi}{25} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{\pi}{18} \end{pmatrix}$$

```
Table[KroneckerDelta[m, n] 2π/n^2, {m, 1, 6}, {n, 1, 6}] // MatrixForm
```

$$\begin{pmatrix} 2\pi & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{\pi}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{2\pi}{9} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{\pi}{8} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{2\pi}{25} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{\pi}{18} \end{pmatrix}$$

But *Mathematica* was unable in 30 minutes to say anything about the integral

$$\int_{-1}^{1} \frac{1}{\sqrt{1-x^2}} \mathrm{Log}\left[1+h^2-2\,h\,x\right] \mathrm{Log}\left[1+g^2-2\,g\,x\right] \, dx$$

which in view of the preceding table we expect to have the value

$$\sum_{k=1}^{\infty} \frac{2\pi}{k^2} gh^k$$

```
2 π PolyLog[2, gh]
```

Which it might be interesting to check out numerically…but I won't.

■ Sabber polynomials

■ Polynomials of the 1st kind

The generating function is

```
In[201]:= S = (1 - 3 x h - h^3)^-ν
Out[201]= (1 - h^3 - 3 h x)^-ν
In[202]:= SabberSeries1 = Series[S, {h, 0, 10}];
In[203]:= Table[{n, C_n = SeriesCoefficient[SabberSeries1, n]}, {n, 0, 5}] // TableForm
Out[203]//TableForm=
```

0	1
1	$3\,x\,\nu$
2	$\frac{9}{2} x^2 \nu\,(1+\nu)$
3	$\nu + \frac{9}{2} x^3 (-2-\nu)\,(-1-\nu)\,\nu$
4	$-3\,x\,(-1-\nu)\,\nu - \frac{27}{8} x^4 (-3-\nu)\,(-2-\nu)\,(-1-\nu)\,\nu$
5	$\frac{9}{2} x^2 (-2-\nu)\,(-1-\nu)\,\nu + \frac{81}{40} x^5 (-4-\nu)\,(-3-\nu)\,(-2-\nu)\,(-1-\nu)\,\nu$

I extend the table (but don't print it):

```
Table[{n, C_n = SeriesCoefficient[SabberSeries1, n]}, {n, 0, 10}];
```

Sabber asserts that these are solutions of the following 3rd order DE:

```
SDE1[f_, n_] := (4 x^3 + 1) D[f, {x, 3}] + 6 (3 + 2 ν) x^2 D[f, {x, 2}] -
   (3 n^2 + 3 n (2 ν + 1) - (3 ν + 2) (3 ν + 5)) x D[f, {x, 1}] - n (n + 3 ν) (n + 3 ν + 3) f
```

```
Table[{n, Simplify[SDE1[𝒞n, n]]}, {n, 0, 10}] // TableForm
```

0	0
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

Splendid! The polynomials $\mathcal{C}_n$ do satisfy that DE. QUESTION: By what argument was Ahmed **led** to that DE? And why is this a 3rd-order DE. My experience + the immediately available literature lead me to expect a DE of 2nd order.

- **Polynomials of the 2nd kind**

The generating function is

In[261]:= `T = (1 + 3 x h^2 - h^3)^-ν`

Out[261]= $\left(1 - h^3 + 3 h^2 x\right)^{-\nu}$

In[262]:= `SabberSeries2 = Series[T, {h, 0, 10}];`

In[263]:= `Table[{n, 𝒟n = SeriesCoefficient[SabberSeries2, n]}, {n, 0, 8}] // TableForm`

Out[263]//TableForm=

0	1
1	0
2	$-3 x \nu$
3	ν
4	$\frac{9}{2} x^2 \nu (1+\nu)$
5	$-3 x \nu (1+\nu)$
6	$-\frac{1}{2} (-1-\nu) \nu - \frac{9}{2} x^3 (-2-\nu) (-1-\nu) \nu$
7	$\frac{9}{2} x^2 \nu (1+\nu) (2+\nu)$
8	$-\frac{3}{2} x (-2-\nu) (-1-\nu) \nu - \frac{27}{8} x^4 (-3-\nu) (-2-\nu) (-1-\nu) \nu$

NOTE: *Linear independence has been lost*; I once, in an early letter, wrote about this. These polynomials do not look interesting-/promising.

```
Table[{n, 𝒟n = SeriesCoefficient[SabberSeries2, n]}, {n, 0, 10}];
```

IMPORTANT NOTE: In the draft of this material that I sent to Ahmed on 10 October I had mistakenly written **`SabberSeries1`** into that command, which led me to think that the $\mathcal{D}$ polynomials did not satisfy the following DE.

Substitutional sense in which the DEs are siblings

Ahmed claims that those polynomials satisfy the following 3rd-order DE:

```
SDE2[f_, n_] := (4 x^3 + 1) D[f, {x, 3}] + 6 (3 + 2 ν) x^2 D[f, {x, 2}] -
   (3 n^2 - 3 n - 10 - 30 ν + 12 n ν) x D[f, {x, 1}] + n (n + 3 ν) (n - 3) f
```

I repeat, for purposes of comparison, the previous differential operator:

```
SDE1[f_, n_] := (4 x^3 + 1) D[f, {x, 3}] + 6 (3 + 2 ν) x^2 D[f, {x, 2}] -
   (3 n^2 + 3 n (2 ν + 1) - (3 ν + 2) (3 ν + 5)) x D[f, {x, 1}] - n (n + 3 ν) (n + 3 ν + 3) f
```

We demonatrate that (as Ahmed has observed) n → -n - 3 ν does indeed send SDE2 → SDE1. The operators differ only in the coefficients that contain n. Looking first to the coefficients of f '

```
- (3 n^2 - 3 n - 10 - 30 ν + 12 n ν) /. n → -n - 3 ν // Simplify
- (3 n^2 + 3 n (2 ν + 1) - (3 ν + 2) (3 ν + 5)) // Simplify
```

$10 - 3 n^2 + 21 \nu + 9 \nu^2 - 3 n (1 + 2 \nu)$

$10 - 3 n^2 + 21 \nu + 9 \nu^2 - 3 n (1 + 2 \nu)$

we have agreement. Looking finally to the coefficients of f

```
n (n + 3 ν) (n - 3) /. n → -n - 3 ν // Simplify
-n (n + 3 ν) (n + 3 ν + 3) // Simplify
```

$-n (n + 3 \nu) (3 + n + 3 \nu)$

$-n (n + 3 \nu) (3 + n + 3 \nu)$

we again have agreement. So Ahmed's claim re the relationship between the two operators is correct.

- **Correction of a COPY-PASTE error**

But the $\mathcal{D}$ polynomials do not satisfy the equations SDE2. So I originally stated. But by corrrection of the previously noted error I find that they DO satisfy the DE:

```
Table[{n, Simplify[SDE2[𝒟n, n]]}, {n, 0, 10}] // TableForm
```

0	0
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

I sent a pdf version of the above material to Ahmed, who responded that "The MYSTERY for me remains the possible link between the C polynomials and the $\mathcal{D}$ polynomials." I remain persuaded that since the C polynomials are manifestly linearly independent, while the $\mathcal{D}$ polynomials are manifestly linearly interdependent, there can exist no such unique and interesting link.

NOTE: The following material was written after the original pdf was written and delivered to Ahmed.

A second pair of Sabber polynomials

These alternative polynomials stand to the original Sabber polynomials rather like the Gegenbauer polynomials stand to the Legendre polynomials.

Alternative polynomials of the 1st kind

The generating function

$$S = \left(1 - 3\,x\,h - h^3\right)^{-\nu}$$

is replaced by

In[120]:=
```
S2 = Log[1 - 3 x h - h^3]
```

Out[120]= $\text{Log}\left[1 - h^3 - 3\,h\,x\right]$

from which the ν-parameter has disappeared (has become an inconsequential factor).

In[121]:=
```
AlternativeSabberSeries1 = Series[S2, {h, 0, 10}];
```

In[122]:=
```
Table[{n, 𝒫n = SeriesCoefficient[AlternativeSabberSeries1, n]}, {n, 0, 10}] // TableForm
```

Out[122]//TableForm=

0	0
1	$-3\,x$
2	$-\frac{9\,x^2}{2}$
3	$-1 - 9\,x^3$
4	$-3\,x - \frac{81\,x^4}{4}$
5	$-9\,x^2 - \frac{243\,x^5}{5}$
6	$-\frac{1}{2} - 27\,x^3 - \frac{243\,x^6}{2}$
7	$-3\,x - 81\,x^4 - \frac{2187\,x^7}{7}$
8	$-\frac{27\,x^2}{2} - 243\,x^5 - \frac{6561\,x^8}{8}$
9	$-\frac{1}{3} - 54\,x^3 - 729\,x^6 - 2187\,x^9$
10	$-3\,x - \frac{405\,x^4}{2} - 2187\,x^7 - \frac{59\,049\,x^{10}}{10}$

Following Ahmed, we construct the differential operator

```
AltSDE1[f_, n_] :=
  (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 + 3 n - 10) x D[f, {x, 1}] - n^2 (n + 3) f
```

```
Table[{n, Simplify[AltSDE1[𝒫n, n]]}, {n, 0, 10}] // TableForm
```

0	0
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

So the (linearly independent) $\mathcal{P}$ polynomials do satisfy that DE.

- **Alternative polynomials of the 2nd kind**

The generating function

$$T = \left(1 + 3\,x\,h^2 - h^3\right)^{-\nu}$$

is replaced by

In[156]:= `T2 = Log[1 + 3 x h^2 - h^3]`

Out[156]= $\text{Log}\left[1 - h^3 + 3\,h^2\,x\right]$

In[157]:= `AlternativeSabberSeries2 = Series[T2, {h, 0, 10}];`

In[158]:= `Table[{n, Qn = SeriesCoefficient[AlternativeSabberSeries2, n]}, {n, 0, 10}] // TableForm`

Out[158]//TableForm=

0	0
1	0
2	$3\,x$
3	-1
4	$-\frac{9\,x^2}{2}$
5	$3\,x$
6	$-\frac{1}{2} + 9\,x^3$
7	$-9\,x^2$
8	$3\,x - \frac{81\,x^4}{4}$
9	$-\frac{1}{3} + 27\,x^3$
10	$-\frac{27\,x^2}{2} + \frac{243\,x^5}{5}$

NOTE that linear independence has again been lost.

```
AltSDE2[f_, n_] :=
  (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 - 3 n - 10) x D[f, {x, 1}] + n^2 (n - 3) f
```

```
Table[{n, Simplify[AltSDE2[Qn, n]]}, {n, 0, 10}] // TableForm
```

0	0
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

Again, the Q polynomials do satisfy the DE proposed by Ahmed.

- **Substitutional sense in which that alternate DEs are siblings**

```
AltSDE2[f_, n_] :=
  (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 - 3 n - 10) x D[f, {x, 1}] + n^2 (n - 3) f

AltSDE1[f_, n_] :=
  (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 + 3 n - 10) x D[f, {x, 1}] - n^2 (n + 3) f

(3 n^2 - 3 n - 10) /. n → -n // Simplify
Simplify[(3 n^2 + 3 n - 10)]
```

$-10 + 3 n + 3 n^2$

$-10 + 3 n + 3 n^2$

```
n^2 (n - 3) /. n → -n // Simplify
Simplify[-n^2 (n + 3)]
```

$-n^2 (3 + n)$

$-n^2 (3 + n)$

So n → -n sends AltSDE2 → AltSDE1.

General solutions of the Sebbar DEs

In the preceding discussion, patterns are most evident not in the polynomials but in the DEs that they satisfy. All four of Armand's DEs are of the form

$$(4 x^3 + 1) f'''[x] + a x^2 f''[x] + b x f'[x] + c f[x] == 0$$

In response to the command

```
DSolve[(4 x^3 + 1) f'''[x] + a x^2 f''[x] + b x f'[x] + c f[x] == 0, f[x], x]
```

Mathematica produces a result of the form

$$f[x] = C_1\,\text{HypergeometricPFQ}\Big[\{a_1, b_1, c_1\}, \Big\{\frac{1}{3}, \frac{2}{3}\Big\}, -4x^3\Big] + x\,C_2\,\text{HypergeometricPFQ}\Big[\{a_2, b_2, c_2\}, \Big\{\frac{2}{3}, \frac{4}{3}\Big\}, -4x^3\Big] + x^2\,C_3\,\text{HypergeometricPFQ}\Big[\{a_3, b_3, c_3\}, \Big\{\frac{4}{3}, \frac{5}{3}\Big\}, -4x^3\Big]$$

where $\{a_1, b_1, c_1, a_2, b_2, c_2, a_3, b_3, c_3\}$ are enormously comlicated functions of the parameters a, b, c (they run—after simplification—to many pages even when printed at 75%). To obtain useful information we must sharpen the focus of this discussion.

- **Solution of the generic DE for $\mathcal{P}$ polynomials**

I look to $\mathcal{P}$ polynomials because they are manifestly linearly independent ($\mathcal{P}_n$ is of order n in x) and because they are distinguished by a single parameter (namely n). The generic differential operator in this case was

```
AltSDE1[f_, n_] :=
 (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 + 3 n - 10) x D[f, {x, 1}] - n^2 (n + 3) f
```

so we look to *Mathematica*'s response to the following command:

```
In[77]:= DSolve[(4 x^3 + 1) f'''[x] + 18 x^2 f''[x] - (3 n^2 + 3 n - 10) x f'[x] - n^2 (n + 3) f[x] == 0, f[x], x]
```

Out[77]=
$$\Big\{\Big\{f[x] \to 2^{2/3}\,x\,C[2]\,\text{HypergeometricPFQ}\Big[\Big\{\frac{1}{3}-\frac{n}{3}, \frac{1}{3}+\frac{n}{6}, \frac{5}{6}+\frac{n}{6}\Big\}, \Big\{\frac{2}{3}, \frac{4}{3}\Big\}, -4x^3\Big] + 2\,2^{1/3}\,x^2\,C[3]\,\text{HypergeometricPFQ}\Big[\Big\{\frac{2}{3}-\frac{n}{3}, \frac{2}{3}+\frac{n}{6}, \frac{7}{6}+\frac{n}{6}\Big\}, \Big\{\frac{4}{3}, \frac{5}{3}\Big\}, -4x^3\Big] + C[1]\,\text{HypergeometricPFQ}\Big[\Big\{\frac{1}{2}+\frac{n}{6}, -\frac{n}{3}, \frac{n}{6}\Big\}, \Big\{\frac{1}{3}, \frac{2}{3}\Big\}, -4x^3\Big]\Big\}\Big\}$$

```
In[118]:= ℙ = 2^(2/3) x C[2] HypergeometricPFQ[{1/3 - n/3, 1/3 + n/6, 5/6 + n/6}, {2/3, 4/3}, -4 x^3] +
   2 2^(1/3) x^2 C[3] HypergeometricPFQ[{2/3 - n/3, 2/3 + n/6, 7/6 + n/6}, {4/3, 5/3}, -4 x^3] +
   C[1] HypergeometricPFQ[{1/2 + n/6, -n/3, n/6}, {1/3, 2/3}, -4 x^3];
```

- **The case n = 0**

```
ℙ /. n → 0
```

Out[119]=
$$C[1] + \frac{2\,x\,C[2]\,\text{HypergeometricPFQ}\Big[\Big\{\frac{1}{3}, \frac{1}{3}\Big\}, \Big\{\frac{4}{3}\Big\}, \frac{1}{2}-\frac{1}{2}\sqrt{1+4x^3}\Big]}{\left(1+\sqrt{1+4x^3}\right)^{1/3}} + \frac{4\,x^2\,C[3]\,\text{HypergeometricPFQ}\Big[\Big\{\frac{2}{3}, \frac{2}{3}\Big\}, \Big\{\frac{5}{3}\Big\}, \frac{1}{2}-\frac{1}{2}\sqrt{1+4x^3}\Big]}{\left(1+\sqrt{1+4x^3}\right)^{2/3}}$$

Compare

In[124]:= $\mathcal{P}_0$

Out[124]= 0

- **The case n = 1**

In[125]:= **ℙ /. n → 1**

Out[125]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}\right\}, \left\{\frac{1}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{5}{6}\right\}, \left\{\frac{5}{3}\right\}, -4 x^3\right]$

Compare

In[126]:= $\mathcal{P}_1$

Out[126]= $-3\, x$

- **The case n = 2**

In[127]:= **ℙ /. n → 2**

Out[127]= $2\, 2^{1/3}\, x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{5}{6}\right\}, \left\{\frac{2}{3}\right\}, -4 x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{7}{6}\right\}, \left\{\frac{4}{3}\right\}, -4 x^3\right]$

Compare

In[128]:= $\mathcal{P}_2$

Out[128]= $-\frac{9 x^2}{2}$

- **The case n = 3**

In[129]:= **ℙ /. n → 3**

Out[129]= $\left(1 + 9 x^3\right) C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{5}{6}\right\}, \left\{\frac{2}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{7}{6}\right\}, \left\{\frac{4}{3}\right\}, -4 x^3\right]$

Compare

In[130]:= $\mathcal{P}_3$

Out[130]= $-1 - 9\, x^3$

- **The case n = 4**

In[131]:= **ℙ /. n → 4**

Out[131]= $2^{2/3}\, x \left(1 + \frac{27 x^3}{4}\right) C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{4}{3}, \frac{7}{6}\right\}, \left\{\frac{1}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{11}{6}\right\}, \left\{\frac{5}{3}\right\}, -4 x^3\right]$

In[135]:= **Together**$\left[x\left(1+\frac{27 x^3}{4}\right)\right]$

Out[135]= $\frac{1}{4} x \left(4+27 x^3\right)$

Compare

In[133]:= $\mathcal{P}_4$ **// Simplify**

Out[133]= $-\frac{3}{4} x \left(4+27 x^3\right)$

- **The case n = 5**

In[136]:= $\mathbb{P}$ **/. n → 5**

Out[136]= $2\, 2^{1/3} x^2 \left(1+\frac{27 x^3}{5}\right) C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{3}, \frac{5}{6}, \frac{4}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] +$
$2^{2/3} x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{4}{3}, \frac{7}{6}, \frac{5}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4 x^3\right]$

In[139]:= **Together**$\left[x^2\left(1+\frac{27 x^3}{5}\right)\right]$

Out[139]= $\frac{1}{5} x^2 \left(5+27 x^3\right)$

Compare

In[138]:= $\mathcal{P}_5$ **// Simplify**

Out[138]= $-\frac{9}{5} x^2 \left(5+27 x^3\right)$

- **The case n = 6**

In[140]:= $\mathbb{P}$ **/. n → 6**

Out[140]= $\left(1+54 x^3+243 x^6\right) C[1] + 2^{2/3} x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{3}, \frac{11}{6}\right\}, \left\{\frac{2}{3}\right\}, -4 x^3\right] +$
$2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{4}{3}, \frac{13}{6}\right\}, \left\{\frac{4}{3}\right\}, -4 x^3\right]$

Compare

In[142]:= $\mathcal{P}_6$ **// Together**

Out[142]= $\frac{1}{2} \left(-1-54 x^3-243 x^6\right)$

So it goes. We see that whenever n is a natural number the general solution of the generic DE assumes the form

(constant) $\mathcal{P}_m$ + Hypergeometric + Hypergeometric

—precisely as Ahmed remarked during the first moments of our Sunday excursion (though how he could know this—since he works by hand, without the assistance of a resource like *Mathematica*—mystifies me!).

The general solution is a sum of three hypergeometric functions

$\mathbb{P} = C_1$ Hypergeometric + C_2 Hypergeometric + C_3 Hypergeometric

Remarkably, as n ranges on the natural numbers it is sometimes one, sometimes another of the hypergometric terms that produces the polynomial. This would appear to close the door to any search for pattern among the $\mathcal{P}$ polynomials. This aspect of the theory evidently hinges on deep properties of generalized hypergeometric functions.

■ **Solution of the generic DE for $\mathcal{Q}$ polynomials**

Proceding as before

```
AltSDE2[f_, n_] :=
  (4 x^3 + 1) D[f, {x, 3}] + 18 x^2 D[f, {x, 2}] - (3 n^2 - 3 n - 10) x D[f, {x, 1}] + n^2 (n - 3) f
```

```
In[145]:= DSolve[(4 x^3 + 1) f'''[x] + 18 x^2 f''[x] - (3 n^2 - 3 n - 10) x f'[x] + n^2 (n - 3) f[x] == 0, f[x], x]
```

Out[145]= $\left\{\left\{f[x] \to 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]\right\}\right\}$

In[148]:= $\mathbb{Q} = 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right];$

But this is precisely the functions obtained from $\mathbb{P}$ by negating the sign of n:

```
In[146]:= ℙ /. n → -n
```

Out[146]= $2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]$

So $\mathbb{P}$ produces $\mathcal{P}$ polynomials as n ranges on the natural numbers, and produces $\mathcal{Q}$ polynomials as n ranges on the **negated** natural numbers.

● **The case n = 0**

In[159]:= 𝕼 /. n → 0
Q_0

Out[159]= $$C[1] + \frac{2\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{1}{3}\right\}, \left\{\frac{4}{3}\right\}, \frac{1}{2} - \frac{1}{2}\sqrt{1 + 4\, x^3}\right]}{\left(1 + \sqrt{1 + 4\, x^3}\right)^{1/3}} + \frac{4\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}, \frac{2}{3}\right\}, \left\{\frac{5}{3}\right\}, \frac{1}{2} - \frac{1}{2}\sqrt{1 + 4\, x^3}\right]}{\left(1 + \sqrt{1 + 4\, x^3}\right)^{2/3}}$$

Out[160]= 0

- **The case n = 1**

In[195]:= 𝕼 /. n → 1
ℙ /. n → -1
Q_1

Out[195]= $$C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}\right\}, \left\{\frac{2}{3}\right\}, -4\, x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{6}, \frac{2}{3}\right\}, \left\{\frac{4}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}, 1, 1\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$$

Out[196]= $$C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}\right\}, \left\{\frac{2}{3}\right\}, -4\, x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{6}, \frac{2}{3}\right\}, \left\{\frac{4}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}, 1, 1\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$$

Out[197]= 0

NOTE: This case appears to be exceptional. It is not obvious that constants C[i] (i = 1, 2, 3) can be found that from that series of hypergeometric functions reproduces the 0-polynomial.

- **The case n = 2**

In[177]:=
```
ℚ /. n → 2
ℙ /. n → -2
Q₂
```

Out[177]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}\right\}, \left\{\frac{1}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{5}{6}\right\}, \left\{\frac{5}{3}\right\}, -4 x^3\right]$

Out[178]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}\right\}, \left\{\frac{1}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{5}{6}\right\}, \left\{\frac{5}{3}\right\}, -4 x^3\right]$

Out[179]= $3\, x$

- **The case n = 3**

In[180]:=
```
ℚ /. n → 3
ℙ /. n → -3
Q₃
```

Out[180]= $C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}\right\}, \left\{\frac{2}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{6}, \frac{2}{3}\right\}, \left\{\frac{4}{3}\right\}, -4 x^3\right]$

Out[181]= $C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}\right\}, \left\{\frac{2}{3}\right\}, -4 x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{6}, \frac{2}{3}\right\}, \left\{\frac{4}{3}\right\}, -4 x^3\right]$

Out[182]= -1

- **The case n = 4**

In[183]:=
```
ℚ /. n → 4
ℙ /. n → -4
Q₄
```

Out[183]= $2\, 2^{1/3} x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, -\frac{1}{6}, \frac{4}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}, \frac{5}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4 x^3\right]$

Out[184]= $2\, 2^{1/3} x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, -\frac{1}{6}, \frac{4}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}, \frac{5}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4 x^3\right]$

Out[185]= $-\frac{9 x^2}{2}$

- **The case n = 5**

In[186]:= **ℚ /. n → 5**
ℙ /. n → -5
$\mathcal{Q}_5$

Out[186]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{6}, -\frac{1}{3}, \frac{5}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}, \frac{7}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$

Out[187]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{6}, -\frac{1}{3}, \frac{5}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}, \frac{7}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$

Out[188]= $3\, x$

- **The case n = 6**

In[189]:= **ℚ /. n → 6**
ℙ /. n → -6
$\mathcal{Q}_6$ // Together

Out[189]= $\left(1 - 18\, x^3\right) C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, -\frac{1}{6}, \frac{7}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}, \frac{8}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$

Out[190]= $\left(1 - 18\, x^3\right) C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, -\frac{1}{6}, \frac{7}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}, \frac{8}{3}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$

Out[191]= $\frac{1}{2}\left(-1 + 18\, x^3\right)$

- **The case n = 7**

In[192]:= **ℚ /. n → 7**
ℙ /. n → -7
$\mathcal{Q}_7$ // Together

Out[192]= $2\, 2^{1/3}\, x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{7}{6}, -\frac{2}{3}, \frac{7}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4\, x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{6}, -\frac{1}{3}, \frac{8}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right]$

Out[193]= $2\, 2^{1/3}\, x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{7}{6}, -\frac{2}{3}, \frac{7}{3}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4\, x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{6}, -\frac{1}{3}, \frac{8}{3}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right]$

Out[194]= $-9\, x^2$

So in all cases except the case n = 1 the polynomial fragment is proportional to a $\mathcal{Q}$-polynomial.

Solution of the generic DE for *C* polynomials

Heere a second parameter ν comes into play. Working from

In[240]:=
```
SDE1[f_, n_] := (4 x^3 + 1) D[f, {x, 3}] + 6 (3 + 2 ν) x^2 D[f, {x, 2}] -
  (3 n^2 + 3 n (2 ν + 1) - (3 ν + 2) (3 ν + 5)) x D[f, {x, 1}] - n (n + 3 ν) (n + 3 ν + 3) f
```

we command

In[198]:=
```
DSolve[(4 x^3 + 1) f'''[x] + 6 (3 + 2 ν) x^2 f''[x] -
   (3 n^2 + 3 n (2 ν + 1) - (3 ν + 2) (3 ν + 5)) x f'[x] - n (n + 3 ν) (n + 3 ν + 3) f[x] == 0, f[x], x]
```

Out[198]=
$$\left\{\left\{f[x] \to 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{3}, \frac{1}{3}+\frac{n}{6}+\frac{\nu}{2}, \frac{5}{6}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{3}, \frac{2}{3}+\frac{n}{6}+\frac{\nu}{2}, \frac{7}{6}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{n}{3}, \frac{n}{6}+\frac{\nu}{2}, \frac{1}{2}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right]\right\}\right\}$$

and so have the generic solution

In[200]:=
$$\mathbb{C} = 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{3}, \frac{1}{3}+\frac{n}{6}+\frac{\nu}{2}, \frac{5}{6}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{3}, \frac{2}{3}+\frac{n}{6}+\frac{\nu}{2}, \frac{7}{6}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{n}{3}, \frac{n}{6}+\frac{\nu}{2}, \frac{1}{2}+\frac{n}{6}+\frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right];$$

We let n march upward through the natural numbers, while leaving ν unspecified:

- **The case n = 0**

In[204]:=
```
ℂ /. n → 0
C0
```

Out[204]=
$$C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{1}{3}+\frac{\nu}{2}, \frac{5}{6}+\frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}, \frac{2}{3}+\frac{\nu}{2}, \frac{7}{6}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]$$

Out[205]= 1

- **The case n = 1**

In[206]:=
```
ℂ /. n → 1
C1
```

Out[206]=
$$2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{1}{6}+\frac{\nu}{2}, \frac{2}{3}+\frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}, \frac{5}{6}+\frac{\nu}{2}, \frac{4}{3}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]$$

Out[207]= $3\, x\, \nu$

Here the polynomial supplied by the general solution differs from $C_1$by a factor, which can make no difference since the DE is linear:

In[231]:=
```
SDE1[x, 1] // Simplify
SDE1[3 x ν, 1] // Simplify
```

Out[231]= 0

Out[232]= 0

- **The case n = 2**

In[208]:=
```
ℂ /. n → 2
C2
```

Out[208]= $2\, 2^{1/3}\, x^2\, C[3] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{1}{3}+\frac{\nu}{2}, \frac{5}{6}+\frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4\, x^3\right] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{2}{3}+\frac{\nu}{2}, \frac{7}{6}+\frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right]$

Out[209]= $\frac{9}{2}\, x^2\, \nu\, (1+\nu)$

The same remark pertains:

In[233]:=
```
SDE1[x^2, 2] // Simplify
SDE1[9/2 x^2 ν (1 + ν), 2] // Simplify
```

Out[233]= 0

Out[234]= 0

- **The case n = 3**

In[210]:=
```
ℂ /. n → 3
C3
```

Out[210]= $\left(1 + 18\, x^3 \left(\frac{1}{2}+\frac{\nu}{2}\right)\left(1+\frac{\nu}{2}\right)\right) C[1] + 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{5}{6}+\frac{\nu}{2}, \frac{4}{3}+\frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4\, x^3\right] + 2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{3}, \frac{7}{6}+\frac{\nu}{2}, \frac{5}{3}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4\, x^3\right]$

Out[211]= $\nu + \frac{9}{2}\, x^3\, (-2-\nu)\, (-1-\nu)\, \nu$

Here the polynomial supplied by the general solution looks quite different from C_3, but the former can be written

In[238]:=
```
(1 + 18 x^3 (1/2 + ν/2) (1 + ν/2)) // Expand // Together
```

Out[238]= $\frac{1}{2}\left(2 + 18\, x^3 + 27\, x^3\, \nu + 9\, x^3\, \nu^2\right)$

and the latter can be written

In[239]:= $\nu + \frac{9}{2} x^3 (-2-\nu)(-1-\nu)\nu$ // Expand // Together

Out[239]= $\frac{1}{2}\left(2\nu + 18 x^3 \nu + 27 x^3 \nu^2 + 9 x^3 \nu^3\right)$

and those are seen to differ only by a ν-factor, so both satisfy the DE:

In[235]:= SDE1$\left[\left(1 + 18 x^3 \left(\frac{1}{2}+\frac{\nu}{2}\right)\left(1+\frac{\nu}{2}\right)\right), 3\right]$ // Simplify

SDE1$\left[\nu + \frac{9}{2} x^3 (-2-\nu)(-1-\nu)\nu, 3\right]$ // Simplify

Out[235]= 0

Out[236]= 0

- **The case n = 4**

In[247]:= ℂ /. n → 4

C_4 // Together

Out[247]= $2^{2/3} x \left(1 + \frac{9}{2} x^3 \left(1+\frac{\nu}{2}\right)\left(\frac{3}{2}+\frac{\nu}{2}\right)\right) C[2] +$

$C[1]\,\text{HypergeometricPFQ}\left[\left\{-\frac{4}{3}, \frac{2}{3}+\frac{\nu}{2}, \frac{7}{6}+\frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] +$

$2\, 2^{1/3} x^2 C[3]\,\text{HypergeometricPFQ}\left[\left\{-\frac{2}{3}, \frac{4}{3}+\frac{\nu}{2}, \frac{11}{6}+\frac{\nu}{2}\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4 x^3\right]$

Out[248]= $\frac{3}{8}\left(8 x \nu + 54 x^4 \nu + 8 x \nu^2 + 99 x^4 \nu^2 + 54 x^4 \nu^3 + 9 x^4 \nu^4\right)$

Here the polynomial supplied by the general solution—which can be written

In[221]:= $x\left(1 + \frac{9}{2} x^3 \left(1+\frac{\nu}{2}\right)\left(\frac{3}{2}+\frac{\nu}{2}\right)\right)$ // Expand // Together

Out[221]= $\frac{1}{8}\left(8 x + 54 x^4 + 45 x^4 \nu + 9 x^4 \nu^2\right)$

—and C_4 look quite dirrerent, but both satisfy the DE:

In[243]:= SDE1$\left[\left(8 x \nu + 54 x^4 \nu + 8 x \nu^2 + 99 x^4 \nu^2 + 54 x^4 \nu^3 + 9 x^4 \nu^4\right), 4\right]$ // Simplify

SDE1$\left[\left(8 x + 54 x^4 + 45 x^4 \nu + 9 x^4 \nu^2\right), 4\right]$ // Simplify

Out[243]= 0

Out[244]= 0

This comes about for the familiar reason: they differ by a factor, which I now demonstrate:

In[253]:= Series$\left[\left(8 x \nu + 54 x^4 \nu + 8 x \nu^2 + 99 x^4 \nu^2 + 54 x^4 \nu^3 + 9 x^4 \nu^4\right), \{x, 0, 4\}\right]$ // Normal

$\left(\nu+\nu^2\right)$ Series$\left[\left(8 x + 54 x^4 + 45 x^4 \nu + 9 x^4 \nu^2\right), \{x, 0, 4\}\right]$ // Normal

Out[253]= $8 x\left(\nu+\nu^2\right) + x^4\left(54\nu + 99\nu^2 + 54\nu^3 + 9\nu^4\right)$

Out[254]= $8 x\left(\nu+\nu^2\right) + x^4\left(\nu+\nu^2\right)\left(54 + 45\nu + 9\nu^2\right)$

```
In[256]:= Factor[(54 ν + 99 ν^2 + 54 ν^3 + 9 ν^4)]
          Factor[(ν + ν^2) (54 + 45 ν + 9 ν^2)]

Out[256]= 9 ν (1 + ν) (2 + ν) (3 + ν)

Out[257]= 9 ν (1 + ν) (2 + ν) (3 + ν)
```

- **Solution of the generic DE for $\mathcal{D}$ polynomials**

```
SDE2[f_, n_] := (4 x^3 + 1) D[f, {x, 3}] + 6 (3 + 2 ν) x^2 D[f, {x, 2}] -
   (3 n^2 - 3 n - 10 - 30 ν + 12 n ν) x D[f, {x, 1}] + n (n + 3 ν) (n - 3) f

In[258]:= DSolve[(4 x^3 + 1) f'''[x] + 6 (3 + 2 ν) x^2 f''[x] -
     (3 n^2 - 3 n - 10 - 30 ν + 12 n ν) x f'[x] + n (n + 3 ν) (n - 3) f[x] == 0, f[x], x]
```

Out[258]= $\left\{\left\{f[x] \to 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}+\nu\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]\right\}\right\}$

In[259]:= $\mathbb{D} = 2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}+\nu\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right];$

We see by inspection that $\mathbb{D}$ is produced bya substitutional transformation from $\mathbb{C}$:

```
In[260]:= ℂ /. n → -n - 3 ν
```

Out[260]= $2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{3}-\frac{n}{6}, \frac{5}{6}-\frac{n}{6}, \frac{1}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4x^3\right] + C[1]\, \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}-\frac{n}{6}, -\frac{n}{6}, \frac{n}{3}+\nu\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4x^3\right] + 2\, 2^{1/3} x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{\frac{2}{3}-\frac{n}{6}, \frac{7}{6}-\frac{n}{6}, \frac{2}{3}+\frac{n}{3}+\nu\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4x^3\right]$

The method by which one extracts $\mathcal{D}$-polynomials from $\mathbb{D}$ holds now no mystery, no surprises, so I do not pursue that matter.

On the relation of $\mathcal{C}$-polynomials to $\mathcal{D}$-polynomials

I have in this connection only a solitary observation, which I illustrate by randomly selected example. The following familiar procedure extracts C_5 from the general solution $\mathbb{C}$:

In[264]:=
```
ℂ /. n → 5
```
$\mathcal{C}_5$

Out[264]= $2\, 2^{1/3}\, x^2 \left(1 + \frac{9}{5} x^3 \left(\frac{3}{2} + \frac{\nu}{2}\right) \left(2 + \frac{\nu}{2}\right)\right) C[3] +$

$C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{3}, \frac{5}{6} + \frac{\nu}{2}, \frac{4}{3} + \frac{\nu}{2}\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] +$

$2^{2/3}\, x\, C[2]\, \text{HypergeometricPFQ}\left[\left\{-\frac{4}{3}, \frac{7}{6} + \frac{\nu}{2}, \frac{5}{3} + \frac{\nu}{2}\right\}, \left\{\frac{2}{3}, \frac{4}{3}\right\}, -4 x^3\right]$

Out[265]= $\frac{9}{2} x^2 (-2 - \nu)(-1 - \nu)\, \nu + \frac{81}{40} x^5 (-4 - \nu)(-3 - \nu)(-2 - \nu)(-1 - \nu)\, \nu$

The polynomial is seen in this instance to issue from the C[3] HypergeometricPFQ term. The substitution n → -- n -- 3ν sends ℂ → 𝔻, where in that same order

In[266]:=
```
𝔻 /. n → 5
```
$\mathcal{D}_5$

Out[266]= $2^{2/3}\, x\, C[2] + C[1]\, \text{HypergeometricPFQ}\left[\left\{-\frac{5}{6}, -\frac{1}{3}, \frac{5}{3} + \nu\right\}, \left\{\frac{1}{3}, \frac{2}{3}\right\}, -4 x^3\right] +$

$2\, 2^{1/3}\, x^2\, C[3]\, \text{HypergeometricPFQ}\left[\left\{-\frac{1}{6}, \frac{1}{3}, \frac{7}{3} + \nu\right\}, \left\{\frac{4}{3}, \frac{5}{3}\right\}, -4 x^3\right]$

Out[267]= $-3\, x\, \nu\, (1 + \nu)$

the polynomial is seen to issue from the C[2] HypergeometricPFQ term. Generally, there appears to be no fixed (or obvious) relation between the term that issues $\mathcal{C}_n$and that which issues $\mathcal{D}_n$. The relation Ahmed seeks appears therefore to be buried deep within the theory of generalized hypergeometric functions, and therefore (I anticipate) to be neither interesting nor useful.